Scott Morrison
20003 4814

# Model Solutions for Assignment #3.

12 Jan. 2006

## MATH 800 Assignment #3

2.7.2. Find a Groebner basis for each of the following ideals, w.r.t. (i) lex, (ii) grlex order.

Calculations done in Maple. See attached worksheet.

a) $I = \langle \underbrace{xy-1}_{f_1}, \underbrace{xy^2-x}_{f_2} \rangle$

(i) lex

$G_0 = \{f_1, f_2\}$

$S = S(f_1,f_2) = x^2-y, \quad \overline{S}^{G_0} = x^2-y = f_3$

$G_1 = \{f_1, f_2, f_3\}$

$S = S(f_1,f_3) = y^2-1, \quad \overline{S}^{G_1} = y^2-1 = f_4$

$S = S(f_2,f_3) = -x^2+y^3, \quad \overline{S}^{G_1} = y^3-y = f_5$

$G_2 = \{f_1, f_2, f_3, f_4, f_5\}$

$S = S(f_1,f_4) = x^2-y, \quad \overline{S}^{G_2} = 0$

$S = S(f_2,f_4) = 0, \quad \overline{S}^{G_2} = 0$

$S = S(f_3,f_4) = x^2-y^3, \quad \overline{S}^{G_2} = 0$

$S = S(f_1,f_5) = x^2y-y^2, \quad \overline{S}^{G_2} = 0$

$S = S(f_2,f_5) = 0, \quad \overline{S}^{G_2} = 0$

$S = S(f_3,f_5) = x^2y-y^4, \quad \overline{S}^{G_2} = 0$

$S = S(f_4,f_5) = 0, \quad \overline{S}^{G_2} = 0$

$\therefore G_2 = \{x^2y-1, xy^2-x, x^2-y, y^2-1, y^3-y\}$ is a GB for $I$ w.r.t. $>_{lex}$, $x>y>z$.

(ii) $G_0 = \{f_1, f_2\}$

$S = S(f_1,f_2) = x^2-y, \quad \overline{S}^{G_0} = x^2-y = f_3$

$G_1 = \{f_1, f_2, f_3\}$

$S = S(f_1,f_3) = y^2-1, \quad \overline{S}^{G_1} = y^2-1 = f_4$

$S = S(f_2,f_3) = y^3-x^2, \quad \overline{S}^{G_1} = y^3-y = f_5$

$G_2 = \{f_1, f_2, f_3, f_4, f_5\}$

$S = S(f_1,f_4) = x^2-y, \quad \overline{S}^{G_2} = 0$

$S = S(f_2,f_4) = 0, \quad \overline{S}^{G_2} = 0$

$S = S(f_3,f_4) = -y^3+x^2, \quad \overline{S}^{G_2} = 0$

$S = S(f_1,f_5) = x^2y-y^2, \quad \overline{S}^{G_2} = 0$

12

$S = S(f_4, f_5) = 0, \quad \overline{S}^{G_2} = 0$
$S = S(f_3, f_5) = -y^4 + x^2y, \quad \overline{S}^{G_2} = 0$
$S = S(f_4, f_5) = 0, \quad \overline{S}^{G_2} = 0$

$\therefore G_2 = \{x^2y - 1, xy^2 - x, x^2 - y, y^2 - 1, y^3 - y\}$ is a GB for I w.r.t. $>_{grlex}$, $x > y > z$.
This is the same basis we found for lex order in (i). ✓

b) $I = \langle \underset{f_1}{x^2 + y}, \underset{f_2}{x^4 + 2x^2y + y^2 + 3} \rangle$

i) $G_0 = \{f_1, f_2\}$
$S = S(f_1, f_2) = -x^2y - y^2 - 3, \quad \overline{S}^{G_0} = -3 = f_3$ ✓ ✓
$G_1 = \{f_1, f_2, f_3\}$
$S = S(f_1, f_3) = y, \quad \overline{S}^{G_1} = 0$
$S = S(f_2, f_3) = 2x^2y + y^2 + 3, \quad \overline{S}^{G_1} = 0$
$\therefore G_1 = \{x^2 + y, x^4 + 2x^2y + y^2 + 3, -3\}$ is a GB for I w.r.t. $>_{lex}$, $x > y > z$.

ii) $G_0 = \{f_1, f_2\}$
$S = S(f_1, f_2) = -x^2y - y^2 - 3, \quad \overline{S}^{G_0} = -3 = f_3$
$G_1 = \{f_1, f_2, f_3\}$
$S = S(f_1, f_3) = y, \quad \overline{S}^{G_1} = 0$
$S = S(f_2, f_3) = 2x^2y + y^2 + 3, \quad \overline{S}^{G_1} = 0$
$\therefore G_1 = \{x^2 + y, x^4 + 2x^2y + y^2 + 3, -3\}$ is a GB for I w.r.t. $>_{grlex}$, $x > y > z$.
This is the same basis we found for lex order in (i) ✓

Since the constant $-3 \in G_1 \subset I$, $I = k[x,y]$.
Thus $\mathbb{V}(I) = \emptyset$. ✓

c) $I = \langle \underset{f_1}{x - z^4}, \underset{f_2}{y - z^5} \rangle$

i) $G_0 = \{f_1, f_2\}$
$S = S(f_1, f_2) = xz^5 - yz^4, \quad \overline{S}^{G_0} = 0$ ✓
$\therefore G_0 = \{x - z^4, y - z^5\}$ is a GB for I w.r.t. $>_{lex}$, $x > y > z$. ✓

ii) $G_0 = \{f_1, f_2\}$

$S = S(f_1, f_2) = -xz + y, \quad \overline{S}^{G_0} = -xz + y = f_3$ ✓ ✓

$G_1 = \{f_1, f_2, f_3\}$

$S = S(f_1, f_3) = yz^3 - x^2, \quad \overline{S}^{G_1} = yz^3 - x^2 = f_4$ ✓ ✓

$S = S(f_2, f_3) = yz^4 - xy, \quad \overline{S}^{G_1} = 0$ ✓ ✓

$G_2 = \{f_1, f_2, f_3, f_4\}$

$S = S(f_1, f_4) = x^2z - xy, \quad \overline{S}^{G_2} = 0$ ✓ ✓

$S = S(f_2, f_4) = x^2z^2 - y^2, \quad \overline{S}^{G_2} = 0$ ✓ ✓

$S = S(f_3, f_4) = -y^2z^2 + x^3, \quad \overline{S}^{G_2} = -y^2z^2 + x^3 = f_5$ ✓ ✓

$G_3 = \{f_1, f_2, f_3, f_4, f_5\}$ ✓

$S = S(f_1, f_5) = x^3z^2 - xy^2, \quad \overline{S}^{G_3} = 0$ ✓ ✓

$S = S(f_2, f_5) = x^3z^3 - y^3, \quad \overline{S}^{G_3} = 0$ ✓ ✓

$S = S(f_3, f_5) = x^4 - y^3z, \quad \overline{S}^{G_3} = x^4 - y^3z = f_6$ ✓ ✓

$S = S(f_4, f_5) = x^3z - x^2y, \quad \overline{S}^{G_3} = 0$ ✓ ✓

$G_4 = \{f_1, f_2, f_3, f_4, f_5, f_6\}$

$S = S(f_1, f_6) = y^3z^5 - x^5, \quad \overline{S}^{G_4} = 0$ ✓ ✓

$S = S(f_2, f_6) = y^3z^6 - x^4y, \quad \overline{S}^{G_4} = 0$ ✓

$S = S(f_3, f_6) = y^3z^2 - x^3y, \quad \overline{S}^{G_4} = 0$ ✓

$S = S(f_4, f_6) = y^4z^4 - x^6, \quad \overline{S}^{G_4} = 0$ ✓

$S = S(f_5, f_6) = y^5z^3 - x^7, \quad \overline{S}^{G_4} = 0$ ✓

$\therefore G_4 = \{-z^4 + x, -z^5 + y, -xz + y, yz^3 - x^2, -y^2z^2 + x^3, x^4 - y^3z\}$ is a GB w.r.t. $>_{grlex}$, $x > y > z$. ✓ ✓ ✓ ✓ ✓ ✓

For this ideal, the lex order GB is much simpler than the grlex order GB. This is related to the fact that the lex leading terms are simpler (lower degree) than the grlex leading terms for the original generators $f_1, f_2$ of I.

2.7.3. Find reduced Groebner bases for the ideals in Exercise 2, w.r.t. (i) lex, (ii) grlex order.

6

a) i) $LT(f_1) = x^2y$, $LT(f_2) = xy^2$, $LT(f_5) = -x^2 \in \langle LT(G_2 \setminus \{f_1, f_2, f_5\})\rangle = \langle x^2, y^2 \rangle$ ✓

so $G_2 \setminus \{f_1, f_2, f_5\} = \{f_3, f_4\} = \{x^2 - y, y^2 - 1\}$ is also a GB for I.

Since no term of $f_3$ is in $\langle y^2 \rangle$ and no term of $f_4$ is in $\langle x^2 \rangle$, and $f_3, f_4$ are both monic, $G_3 = \{x^2 - y, y^2 - 1\}$ is a reduced GB for I w.r.t. $>_{lex}$, $x > y > z$ ✓

ii) In Exercise 2(a) the Groebner bases we found for I w.r.t. lex and grlex orders were the same, and (most importantly) the leading terms are the same for both orderings. Therefore $G_3 = \{x^2-y, y^2-1\}$ from part (i) is also a reduced GB for I w.r.t. $>_{grlex}$, $x>y>z$.

b) i) $LT(f_1) = x^2$, $LT(f_2) = x^4 \in \langle LT(G_1 \setminus \{f_1, f_2\})\rangle = \langle -3 \rangle$
so $G_1 \setminus \{f_1, f_2\} = \{f_3\} = \{3\}$ is also a GB for I. ✓
Making all elements monic, $G_2 = \{\frac{1}{3} f_3\} = \{1\}$ is a reduced GB for I w.r.t. $>_{lex}$, $x>y>z$. ✓

ii) In Exercise 2(b) the Groebner bases we found for I w.r.t. lex and grlex orders were the same, with the same leading terms. Therefore $G_2 = \{1\}$ from part (i) is also a reduced GB for I w.r.t. $>_{grlex}$, $x>y>z$. ✓

So $I = \langle 1 \rangle = k[x,y]$, as we stated in Exercise 2. ✓

c) i) $G_0 = \{f_1, f_2\} = \{x-z^4, y-z^5\}$ is a reduced GB for I w.r.t. $>_{lex}$, $x>y>z$, because no term of $x-z^4$ is in $\langle y \rangle$ and no term of $y-z^5$ is in $\langle x \rangle$, and both polynomials are monic. ✓

ii) $LT(f_1) = -z^4 \mid LT(f_2) = -z^5$, so $LT(f_2) \in \langle LT(G_4 \setminus \{f_2\})\rangle$.
so $G_5 = G_4 \setminus \{f_2\} = \{f_1, f_3, f_4, f_5, f_6\} = \{-z^4+x, -xz+y, yz^3-x^2, -y^2z^2+x^3, x^4-y^3z\}$ is also a GB for I. ✓
No term of $f_1 = -z^4+x$ is in $\langle LT(G_5 \setminus \{f_1\})\rangle = \langle -xz, yz^3, -y^2z^2, x^4\rangle$,
so $f_1$ is reduced for $G_5$.
No term of $f_3 = -xz+y$ is in $\langle LT(G_5 \setminus \{f_3\})\rangle = \langle -z^4, yz^3, -y^2z^2, x^4\rangle$,
so $f_3$ is reduced for $G_5$.
No term of $f_4 = yz^3-x^2$ is in $\langle LT(G_5 \setminus \{f_4\})\rangle = \langle -z^4, -xz, -y^2z^2, x^4\rangle$,
so $f_4$ is reduced for $G_5$.
No term of $f_5 = -y^2z^2+x^3$ is in $\langle LT(G_5 \setminus \{f_5\})\rangle = \langle -z^4, -xz, yz^3, x^4\rangle$,
so $f_5$ is reduced for $G_5$.
No term of $f_6 = x^4-y^3z$ is in $\langle LT(G_5 \setminus \{f_6\})\rangle = \langle -z^4, -xz, yz^3, -y^2z^2\rangle$,
so $f_6$ is reduced for $G_5$.
Making all elements monic, $G_6 = \{-f_1, -f_3, f_4, -f_5, f_6\} = \{z^4-x, xz-y, yz^3-x^2, y^2z^2-x^3, x^4-y^3z\}$ ✓ ✓ ✓ ✓ ✓
is a reduced GB for I w.r.t. $>_{grlex}$, $x>y>z$.

2.7.7. Fix a monomial order, and let $G$ and $\tilde{G}$ be minimal Groebner bases for the ideal $I$.

a) Prove that $LT(G) = LT(\tilde{G})$.

Read this out loud. $g_i \in I$ does not $\Rightarrow LM(g_i) = LT(g_i)$.

Proof: Let $g_i \in G$. Then $g_i \in I$

$G$ is minimal $\Rightarrow LM(g_i) = LT(g_i) \in \langle LM(\tilde{G})\rangle = \langle LT(\tilde{G})\rangle = \langle LT(I)\rangle$, because

X $\tilde{G}$ is a Groebner bases, and $G$ minimal $\Rightarrow LC(g) = 1 \ \forall g \in G$. ✓

$\Rightarrow \exists \tilde{g}_j \in \tilde{G}$ s.t. $LM(\tilde{g}_j) \mid LM(g_i)$

$\Rightarrow LT(g_i) = LM(g_i) = m_1 LM(\tilde{g}_j) = m_1 LT(\tilde{g}_j)$ for some monomial $m_1$. ✓

But similarly $\tilde{g}_j \in \tilde{G} \Rightarrow \exists g_k \in G$ s.t. $LT(\tilde{g}_j) = m_2 LT(g_k)$ ✓

for some monomial $m_2$, since $G$ is also a minimal Groebner basis.

Then $LT(g_i) = m_1 m_2 LT(g_k)$, so $LT(g_i) \in \langle LT(g_k)\rangle$.

5/6

If $k \neq i$ then we would have $LT(g_i) \in \langle LT(G \setminus \{g_i\})\rangle$, contradicting the minimality of $G$. ✓

So $k = i$, giving $LT(g_i) = m_1 m_2 LT(g_i) \Rightarrow m_1 = m_2 = 1$. ✓

Therefore $LT(g_i) = m_1 LT(\tilde{g}_j) = LT(\tilde{g}_j)$, so $LT(g_i) \in LT(\tilde{G})$.

Thus $LT(G) \subset LT(\tilde{G})$, and swapping $G$ and $\tilde{G}$ in the above argument shows $LT(\tilde{G}) \subset LT(G)$.

$\therefore LT(G) = LT(\tilde{G})$. $\square$ ✓

b) Conclude that $G$ and $\tilde{G}$ have the same number of elements.

Since $G$ is minimal, no two elements $g_i, g_j \in G$ have the same leading terms (otherwise $LT(g_i) \in \langle LT(G \setminus \{g_i\})\rangle$, a contradiction).

The same is true for $\tilde{G}$.

So $|G| = |LT(G)| = |LT(\tilde{G})| = |\tilde{G}|$, that is, $G$ and $\tilde{G}$ have the same number of elements. $\square$

2.8.5. [See Maple attachment.]

2.8.11. [See Maple attachment.]

3.1.1. Let $I \subset k[x_1,\dots,x_n]$ be an ideal.

a) Prove that $I_\ell = I \cap k[x_{\ell+1},\dots,x_n]$ is an ideal of $k[x_{\ell+1},\dots,x_n]$.

Proof: (i) $0 \in I$ (since $I$ is an ideal) and $0 \in k \subset k[x_{\ell+1},\dots,x_n]$,
so $0 \in I \cap k[x_{\ell+1},\dots,x_n] = I_\ell$.

(ii) Let $f,g \in I_\ell$. Then $f,g \in I$ and $f,g \in k[x_{\ell+1},\dots,x_n]$
$\Rightarrow f+g \in I$ (an ideal) and $f+g \in k[x_{\ell+1},\dots,x_n]$ (a ring - closed under addition)
$\Rightarrow f+g \in I \cap k[x_{\ell+1},\dots,x_n] = I_\ell$.

3

(iii) Let $f \in I_\ell$ and $h \in k[x_{\ell+1},\dots,x_n]$. Then $f \in I$ and $f \in k[x_{\ell+1},\dots,x_n]$
$\Rightarrow h \cdot f \in I$ (an ideal), since $h \in k[x_1,\dots,x_n]$, and
$h \cdot f \in k[x_{\ell+1},\dots,x_n]$ (a ring - closed under multiplication)
$\Rightarrow h \cdot f \in I \cap k[x_{\ell+1},\dots,x_n] = I_\ell$.

$\therefore I_\ell$ is an ideal of $k[x_{\ell+1},\dots,x_n]$. $\square$

b) Prove that $I_{\ell+1} \subset k[x_{\ell+2},\dots,x_n]$ is the first elimination ideal of $I_\ell \subset k[x_{\ell+1},\dots,x_n]$.

Proof: The first elimination ideal of $J = I_\ell \subset k[x_{\ell+1},\dots,x_n]$ is
$J_1 = J \cap k[x_{(\ell+1)+1},\dots,x_n] = I_\ell \cap k[x_{\ell+2},\dots,x_n]$.
$I_{\ell+1} = I \cap k[x_{\ell+2},\dots,x_n]$.

1

Let $f \in J_1 \Rightarrow f \in I_\ell$ and $f \in k[x_{\ell+2},\dots,x_n]$. But $f \in I_\ell \Rightarrow f \in I$,
$\Rightarrow$ so $f \in I \cap k[x_{\ell+2},\dots,x_n] = I_{\ell+1}$. Thus $J_1 \subset I_{\ell+1}$.
Now let $f \in I_{\ell+1} \Rightarrow f \in I$ and $f \in k[x_{\ell+2},\dots,x_n] \subset k[x_{\ell+1},\dots,x_n]$.
But then $f \in I \cap k[x_{\ell+1},\dots,x_n] = I_\ell = J$, so $f \in J \cap k[x_{\ell+2},\dots,x_n] = J_1$.
Thus $I_{\ell+1} \subset J_1$.
$\therefore I_{\ell+1} = J_1$, the first elimination ideal of $J = I_\ell$. $\square$

3.1.2. [See Maple attachment.]

3.1.4. [See Maple attachment.]

* 3.1.5. Prove the generalized Elimination Theorem – If $I$ is an ideal in $k[x_1,\dots,x_n]$ and $G$ is a Groebner basis of $I$ w.r.t. a monomial order of <u>$l$-elimination type</u> (any monomial involving one of $x_1,\dots,x_l$ is greater than all monomials in $k[x_{l+1},\dots,x_n]$), then $G\cap k[x_{l+1},\dots,x_n]$ is a basis of the $l^{th}$ elimination ideal $I\cap k[x_{l+1},\dots,x_n]$.

Proof: Fix $l$ between 0 and $n$. Let $G_l = G\cap k[x_{l+1},\dots,x_n]$ and $I_l = I\cap k[x_{l+1},\dots,x_n]$.
Then $G_l \subset I_l$, so $\langle LT(G_l)\rangle \subset \langle LT(I_l)\rangle$. ✓

To show that $\langle LT(I_l)\rangle \subset \langle LT(G_l)\rangle$, let $f\in I_l \subset I$.
So $LT(f)$ is divisible by $LT(g)$ for some $g\in G$ since $G$ is a Groebner basis of $I$. ✓
Since $f\in I_l \subset k[x_{l+1},\dots,x_n]$, $LT(f)$ and consequently $LT(g)$ involve only the variables $x_{l+1},\dots,x_n$. ✓
Now because our monomial order is of $l$-elimination type, any monomial involving $x_1,\dots,x_l$ is greater than all monomials in $k[x_{l+1},\dots,x_n]$.
So $LT(g)\in k[x_{l+1},\dots,x_n]$ implies $g\in k[x_{l+1},\dots,x_n]$.
Then $g\in G_l$. So $LT(f) \in \langle LT(g)\rangle \subset \langle LT(G_l)\rangle$. ✓
Thus $\langle LT(I_l)\rangle \subset \langle LT(G_l)\rangle$.
$\therefore$ $\langle LT(G_l)\rangle = \langle LT(I_l)\rangle$, that is, $G_l$ is a Groebner basis of $I_l$ □

5

3.1.6. b) Explain how to create a product order that induces grevlex on both $k[x_1,\dots,x_l]$ and $k[x_{l+1},\dots,x_n]$ and show that this order is of $l$-elimination type.

Define the product monomial order $>_{l\text{-grevlex}}$ on $k[x_1,\dots,x_n]$ as follows:
Write any monomial in $k[x_1,\dots,x_n]$ as $y^\alpha z^\beta$, where $y=(x_1,\dots,x_l)$ and $z=(x_{l+1},\dots,x_n)$, and $\alpha\in\mathbb{Z}^l_{\geq 0}$ and $\beta\in\mathbb{Z}^{n-l}_{\geq 0}$. Then define

$$y^\alpha z^\beta >_{l\text{-grevlex}} y^\gamma z^\delta \iff y^\alpha >_{grevlex} y^\gamma,\ \text{or } y^\alpha = y^\gamma \text{ and } z^\beta >_{grevlex} z^\delta.$$ ✓

Suppose we have two monomials $y^\alpha z^\beta, y^\gamma z^\delta \in k[x_1,\dots,x_n]$ that are in variables $x_{l+1},\dots,x_n$ alone. Then $\alpha=\gamma=0\in\mathbb{Z}^l_{\geq 0}$ so $y^\alpha = y^\gamma = 1$.

Thus $y^\alpha z^\beta >_{l\text{-grevlex}} y^\gamma z^\delta \iff z^\beta >_{grevlex} z^\delta$, so $>_{l\text{-grevlex}}$ induces grevlex order on $k[x_{l+1},\dots,x_n]$. Similarly suppose we have two monomials $y^\alpha z^\beta, y^\gamma z^\delta \in k[x_1,\dots,x_n]$ that are in variables $x_1,\dots,x_l$ alone. Then $\beta=\delta=0\in\mathbb{Z}^{n-l}_{\geq 0}$, so $z^\beta=z^\delta=1$. If the monomials are not the same then $y^\alpha\neq y^\gamma$, thus $y^\alpha z^\beta >_{l\text{-grevlex}} y^\gamma z^\delta \iff y^\alpha >_{grevlex} y^\gamma$, so $>_{l\text{-grevlex}}$ induces grevlex order on $k[x_1,\dots,x_l]$. ✓

4

To show that the order $>_{l\text{-grevlex}}$ is of $l$-elimination type, let $y^\alpha z^\beta$ be a monomial involving one of $x_1,\dots,x_l$ and let $y^\gamma z^\delta$ be any monomial in $k[x_{l+1},\dots,x_n]$.
Then $\gamma=0\in\mathbb{Z}^l_{\geq 0}$ and $\alpha\neq 0\in\mathbb{Z}^l_{\geq 0}$. ✓ By Corollary 6 in §2.4,
$\alpha >_{grevlex} \gamma=0$ ✓ (or $y^\alpha >_{grevlex} y^\gamma=1$) ✓
since $>_{grevlex}$ is a monomial order on $k[x_1,\dots,x_l]$.
So $y^\alpha z^\beta >_{l\text{-grevlex}} y^\gamma z^\delta = z^\delta$.
Therefore $>_{l\text{-grevlex}}$ is of $l$-elimination type. ✓

c) If $G$ is a Groebner basis for $I\subset k[x_1,\dots,x_n]$ w.r.t. the product order defined in part (b), explain why $G\cap k[x_{l+1},\dots,x_n]$ is a Groebner basis w.r.t. grevlex.

OK Since the product order $>_{l\text{-grevlex}}$ is of $l$-elimination type (from part (b)), it follows from the generalized Elimination Theorem (exercise 5) that $G\cap k[x_{l+1},\dots,x_n]$ is a Groebner basis of the $l^{th}$ elimination ideal $I_l=I\cap k[x_{l+1},\dots,x_n]$ w.r.t. $>_{l\text{-grevlex}}$. But $I_l\subset k[x_{l+1},\dots,x_n]$, and we already argued (part (b)) that $>_{l\text{-grevlex}}$ induces grevlex order on $k[x_{l+1},\dots,x_n]$. Therefore $G\cap k[x_{l+1},\dots,x_n]$ is a Groebner basis for $I_l$ w.r.t. grevlex.

3.1.7. [See Maple attachment.]

3.3.6.
3.3.8. [ See Maple attachment. ]
3.3.14.

Additional Exercise 1.
Additional Exercise 2. [See Maple attachment.]

Additional Exercise 3. Compute a reduced Groebner basis for the linear system
$S = \{x+y+z=1,\ x-2y-z=2,\ y+2z=5\}$
using (i) lex ordering, $x>y>z$, and (ii) grevlex ordering, $x>y>z$.

The ideal determined by $S$ is $I = \langle f_1 = x+y+z-1,\ f_2 = x-2y-z-2,\ f_3 = y+2z-5 \rangle$

i) $G_0 = \{f_1, f_2, f_3\}$

$S(f_1,f_2) = 3y+2z+1 \longrightarrow_G -4z+16.$ $(\times -\frac{1}{4} \rightarrow)$ let $f_4 = z-4$ ✓

$S(f_1,f_3) \longrightarrow_G 0$ (by Prop 4, §2.9) since $x,y$ are rel. prime

$S(f_2,f_3) \longrightarrow_G 0$ since $x,y$ are rel. prime

$S(f_1,f_4) \longrightarrow_G 0$ since $x,z$ are rel. prime

$S(f_2,f_4) \longrightarrow_G 0$ since $x,z$ are rel. prime

$S(f_3,f_4) \longrightarrow_G 0$ since $y,z$ are rel. prime ✓

So $G_1 = \{f_1,f_2,f_3,f_4\}$ is a GB for $I$. ✓

$LT(f_2) \in \langle LT(G_1 \setminus \{f_2\}) \rangle$ so $G_2 = \{f_1,f_3,f_4\}$ is a minimal GB for $I$.

$f_1 \div \{f_3, f_4\} \rightarrow x$

$f_3 \div \{f_1, f_4\} \rightarrow y+3$ ✓

$f_4 \div \{f_1, f_3\} \rightarrow z-4 = f_4$ ✓

4

$\therefore G_3 = \{x,\ y+3,\ z-4\}$ is the reduced GB for $I$ w.r.t. lex order, $x>y>z$.

ii) The calculations are identical at every step, so $\{x, y+3, z-4\}$ is also the reduced GB for $I$ w.r.t. grevlex order, $x>y>z$. This is because these two monomial orderings induce exactly the same order on all monomials of total degree $\leq 1$ (the only monomials in a linear system): ✓

$x > y > z > 1.$

Thus all leading terms, division results and S-polynomial calculations are exactly the same for both orderings.

(Note that the reduced GB gives the (unique) reduced row-echelon form of $S$ for the variable ordering $x>y>z$: $\{x=0, y=-3, z=4\}$) ✓

Scott Cowan.

**MATH 800 Assignment 3**
(due June 21, 2006, 9:30)

```
> restart;
  X := [x,y,z]:
```

## Implementation of the Division Algorithm

We program the multivariate division algorithm. Our procedure takes as input a polynomial $f$ to divide, an ordered $s$-tuple $[f_1, ..., f_s]$ to divide by, a variable ordering $X$, and a procedure $\mathrm{LT}(g, X)$ that computes the leading term of a polynomial $g$ with respect to a certain monomial order. The output from our procedure is $[[a_1, ..., a_s], r]$ satisfying the conditions given in the division algorithm. The procedure uses a global procedure *monom_divis* that tests whether one monomial term is divisible by another monomial term. If the optional parameter *verbose* = *true* is given, the procedure will print out its intermediate calculations.

```
> DIVIDE := proc( f, fL::list, X::list, LT::procedure )
  local s, a, r, p, t, i, verb;
  global monom_divis;
    verb := false;
    if (nargs > 4) then for t in args[5..-1] do
      if (op(1,t) = verbose) then verb := op(2,t) fi;
    od fi;
    s := nops(fL);
    for i from 1 to s do a[i] := 0 od;
    (p,r) := (f,0);
    while (p <> 0) do
      if (verb) then print('p' = p) fi;
      i := 1;
      while (i <= s) and not(monom_divis( LT(p,X), LT(fL[i],X), X
 )) do i := i+1 od;
      if (i > s) then
        (p,r) := (p-LT(p,X),r+LT(p,X));
        if (verb) then print('r' = r) fi;
      else
        t := LT(p,X)/LT(fL[i],X);
        (p,a[i]) := (p-expand(t*fL[i]),a[i]+t);
        if (verb) then print(`a[`||i||`]` = a[i]) fi;
      fi;
    od;
    return [[seq(a[i], i=1..s)],r];
  end:
```

We write a procedure to compute the multidegree of a polynomial $f$ in variables $X$. The multidegree is dependent on the monomial ordering chosen, so as in the division algorithm we input a procedure to compute leading terms.

bruno. 7.

```
> multideg := proc( f, X::list, LT::procedure )
    if type( f, `+` ) then return multideg( LT(f,X), X, LT ) fi;
    return map2( degree, f, X );
  end:
```

We write a procedure to test if one monomial term $m_1$ is divisible by another monomial term $m_2$ in variables $X$. Note that the property of divisibility is independent of the monomial ordering chosen.

```
> monom_divis := proc( m1, m2, X::list )
  local a, i;
  global multideg;
    if type( m1, `+` ) or type( m2, `+` ) then error "inputs
  should be monomials" fi;
    a := multideg( m1, X ) - multideg( m2, X );
    for i in a do if (i < 0) then return false fi od;
    return true;
  end:
```

We write a procedure to compute leading terms with respect to lexicographic order.

```
> LTlex := proc( f, X::list ) local c, m;
    c := lcoeff( f, X, 'm' );
    return c*m;
  end:
```

We write a procedure to compute leading terms with respect to graded lexicographic order.

```
> LTgrlex := proc( f, X::list ) local d, g, t;
    if not(type( f, `+` )) then return f fi;
    d := max( seq( degree(t), t=f ) );
    g := add( `if`( degree(t) = d, t, 0 ), t=f );
    return LTlex(g,X);
  end:
```

## Implementation of Buchberger's algorithm

We begin by programming a procedure to calculate S-polynomials. This procedure computes $S(f, g)$ with respect to the monomial order defined by the variable ordering $X$, and the leading-term procedure $LT(g, X)$. If the optional parameter *fractionfree = true* is given then the resulting S-polynomial will be scaled so that none of its coefficients are fractions.

```
> S_poly := proc( f, g, X::list, LT )
  local ltf, ltg, a, b, c, lcm, i, fractfree;
    fractfree := false;
    if (nargs > 4) then for c in args[5..-1] do
      if (op(1,c) = fractionfree) then fractfree := op(2,c) fi;
    od fi;
    (ltf,ltg) := (LT(f,X),LT(g,X));
    (a,b) := multideg( ltf, X ),multideg( ltg, X );
    lcm := mul( X[i]^max(a[i],b[i]), i=1..nops(a) );
    if (fractfree) then
```

```
      c := ilcm( lcoeff( ltf, X ), lcoeff( ltg, X ) );
      lcm := c*lcm;
    fi;
    return expand(lcm/ltf*f)-expand(lcm/ltg*g);
  end:
```

We program Buchberger's algorithm as given in lecture and in the textbook. Our procedure takes as input a list of generators $[f_1, \ldots, f_s]$ for an ideal, a variable ordering $X$, and a procedure LT($g$, $X$) that computes leading terms with respect to a certain monomial order. It outputs a Groebner basis $[g_1, \ldots, g_t]$ for the ideal. This Groebner basis will contain no new duplicate polynomials and will preserve the order of the original polynomials. The procedure uses global procedures *S_poly* for computing S-polynomials and *DIVIDE* to perform the multivariate division algorithm. The optional parameters *verbose* and *fractionfree* can be specified true or false to determine if intermediate calculations will be shown and if S-polynomials will be scaled to be fraction free.

```
> Buchberger := proc( F::list, X::list, LT::procedure )
  local G, G_set, f, g, S, r, i, j, k, verb, fractfree;
  global DIVIDE, S_poly;
    (verb,fractfree) := (false,false);
    if (nargs > 3) then for r in args[4..-1] do
      if (op(1,r) = verbose) then verb := op(2,r);
      elif (op(1,r) = fractionfree) then fractfree := op(2,r)
  fi;
    od fi;
    (G,G_set) := (F,{op(F)});
    j := 1;
    while (j <= nops(G)) do
      g := G[j];
      k := nops(G);
      i := 1;
      while (i < j) do
        f := G[i];
        S := S_poly( f, g, X, LT, fractionfree=fractfree );
        r := DIVIDE( S, G[1..k], X, LT );
        if (verb) then print('S'(f,g) = S, rem = r, 'G' = k) fi;
        r := r[-1];
        if (r <> 0) and not(member( r, G_set )) then
          (G,G_set) := ([op(G),r],G_set union {r}) fi;
        i := i+1;
      od;
      j := j+1;
    od;
    return G;
  end:
```

We write a procedure to take any Groebner basis for an ideal with respect to a given monomial order and transform it into the unique reduced Groebner basis for this ideal with respect to the monomial order.

```
> GroebnerReduce := proc( GB::list, X::list, LT::procedure )
  local G, rG, g, r, verb;
  global DIVIDE;
    verb := false;
    if (nargs > 3) then for r in args[4..-1] do
      if (op(1,r) = verbose) then verb := op(2,r) fi;
    od fi;
    (rG,G) := ([],GB);
    while (G <> []) do
      (g,G) := (G[1],G[2..-1]);
      r := DIVIDE( g, [op(rG),op(G)], X, LT );
      if (verb) then print('g' = g, rem = r, 'G' =
  nops(rG)+nops(G)) fi;
      r := r[-1];
      if (r <> 0) then rG := [op(rG),r/lcoeff( LT(r,X), X )] fi;
    od;
    return rG;
  end:
```

# Calculations for 2.7.2 & 2.7.3

## a)

Input the generators for the ideal.

```
> F := [x^2*y-1, x*y^2-x];
```

$$F := [x^2 y - 1, x y^2 - x]$$

Compute a Groebner basis w.r.t. lex order.

```
> G := Buchberger( F, X, LTlex, verbose=true );
```

$$S(x^2 y - 1, x y^2 - x) = -y + x^2, rem = [[0, 0], -y + x^2], G = 2$$

$$S(x^2 y - 1, -y + x^2) = y^2 - 1, rem = [[0, 0, 0], y^2 - 1], G = 3$$

$$S(x y^2 - x, -y + x^2) = -x^2 + y^3, rem = [[0, 0, -1], y^3 - y], G = 3$$

$$S(x^2 y - 1, y^2 - 1) = -y + x^2, rem = [[0, 0, 1, 0, 0], 0], G = 5$$

$$S(x y^2 - x, y^2 - 1) = 0, rem = [[0, 0, 0, 0, 0], 0], G = 5$$

$$S(-y + x^2, y^2 - 1) = -y^3 + x^2, rem = [[0, 0, 1, -y, 0], 0], G = 5$$

$$S(x^2 y - 1, y^3 - y) = -y^2 + x^2 y, rem = [[1, 0, 0, -1, 0], 0], G = 5$$

$$S(x y^2 - x, y^3 - y) = 0, rem = [[0, 0, 0, 0, 0], 0], G = 5$$

$$S(-y + x^2, y^3 - y) = -y^4 + x^2 y, rem = [[1, 0, 0, -y^2 - 1, 0], 0], G = 5$$

$$S(y^2-1, y^3-y) = 0, rem = [[0, 0, 0, 0, 0], 0], G = 5$$

$$G := [x^2 y - 1, x y^2 - x, -y + x^2, y^2 - 1, y^3 - y]$$

Reduce to the reduced Groebner basis w.r.t. lex order.

```
> GroebnerReduce( G, X, LTlex );
```

$$[-y + x^2, y^2 - 1]$$

Repeat for grlex order.

```
> G := Buchberger( F, X, LTgrlex, verbose=true );
```

$$S(x^2 y - 1, x y^2 - x) = -y + x^2, rem = [[0, 0], -y + x^2], G = 2$$

$$S(x^2 y - 1, -y + x^2) = y^2 - 1, rem = [[0, 0, 0], y^2 - 1], G = 3$$

$$S(x y^2 - x, -y + x^2) = -x^2 + y^3, rem = [[0, 0, -1], y^3 - y], G = 3$$

$$S(x^2 y - 1, y^2 - 1) = -y + x^2, rem = [[0, 0, 1, 0, 0], 0], G = 5$$

$$S(x y^2 - x, y^2 - 1) = 0, rem = [[0, 0, 0, 0, 0], 0], G = 5$$

$$S(-y + x^2, y^2 - 1) = -y^3 + x^2, rem = [[0, 0, 1, -y, 0], 0], G = 5$$

$$S(x^2 y - 1, y^3 - y) = -y^2 + x^2 y, rem = [[1, 0, 0, -1, 0], 0], G = 5$$

$$S(x y^2 - x, y^3 - y) = 0, rem = [[0, 0, 0, 0, 0], 0], G = 5$$

$$S(-y + x^2, y^3 - y) = -y^4 + x^2 y, rem = [[1, 0, 0, -y^2 - 1, 0], 0], G = 5$$

$$S(y^2 - 1, y^3 - y) = 0, rem = [[0, 0, 0, 0, 0], 0], G = 5$$

$$G := [x^2 y - 1, x y^2 - x, -y + x^2, y^2 - 1, y^3 - y]$$

```
> GroebnerReduce( G, X, LTgrlex );
```

$$[-y + x^2, y^2 - 1]$$

## b)

Input the generators for the ideal.

```
> F := [x^2+y, x^4+2*x^2*y+y^2+3];
```

$$F := [x^2 + y, x^4 + 2 x^2 y + y^2 + 3]$$

Compute a Groebner basis w.r.t. lex order.

```
> G := Buchberger( F, X, LTlex, verbose=true );
```

$$S(x^2 + y, x^4 + 2 x^2 y + y^2 + 3) = -x^2 y - y^2 - 3, rem = [[-y, 0], \text{-}3], G = 2$$

$$S(x^2 + y, \text{-}3) = y, rem = \left[\left[0, 0, -\frac{y}{3}\right], 0\right], G = 3$$

$$S(x^4 + 2 x^2 y + y^2 + 3, \text{-}3) = 2 x^2 y + y^2 + 3, rem = \left[\left[2 y, 0, \frac{y^2}{3} - 1\right], 0\right], G = 3$$

$$G := [x^2 + y, x^4 + 2 x^2 y + y^2 + 3, \text{-}3]$$

Reduce to the reduced Groebner basis w.r.t. lex order.

```
> GroebnerReduce( G, X, LTlex );
```

$$[1]$$

Repeat for grlex order.

```
> G := Buchberger( F, X, LTgrlex, verbose=true );
```

$$S(x^2+y, x^4+2x^2y+y^2+3) = -x^2y-y^2-3, \mathit{rem} = [[-y, 0], -3], G=2$$

$$S(x^2+y, -3) = y, \mathit{rem} = \left[\left[0, 0, -\frac{y}{3}\right], 0\right], G=3$$

$$S(x^4+2x^2y+y^2+3, -3) = 2x^2y+y^2+3, \mathit{rem} = \left[\left[2y, 0, \frac{y^2}{3}-1\right], 0\right], G=3$$

$$G := [x^2+y, x^4+2x^2y+y^2+3, -3]$$

```
> GroebnerReduce( G, X, LTgrlex );
```

$$[1]$$

## c)

Input the generators for the ideal.

```
> F := [x-z^4, y-z^5];
```

$$F := [x-z^4, y-z^5]$$

Compute a Groebner basis w.r.t. lex order.

```
> G := Buchberger( F, X, LTlex, verbose=true );
```

$$S(x-z^4, y-z^5) = -yz^4+xz^5, \mathit{rem} = [[z^5, -z^4], 0], G=2$$

$$G := [x-z^4, y-z^5]$$

Reduce to the reduced Groebner basis w.r.t. lex order.

```
> GroebnerReduce( G, X, LTlex );
```

$$[x-z^4, y-z^5]$$

Repeat for grlex order.

```
> G := Buchberger( F, X, LTgrlex, verbose=true );
```

$$S(x-z^4, y-z^5) = -zx+y, \mathit{rem} = [[0, 0], -zx+y], G=2$$

$$S(x-z^4, -zx+y) = -x^2+z^3y, \mathit{rem} = [[0, 0, 0], -x^2+z^3y], G=3$$

$$S(y-z^5, -zx+y) = -yx+yz^4, \mathit{rem} = [[-y, 0, 0], 0], G=3$$

$$S(x-z^4, -x^2+z^3y) = -yx+zx^2, \mathit{rem} = [[0, 0, -x, 0], 0], G=4$$

$$S(y-z^5, -x^2+z^3y) = -y^2+z^2x^2, \mathit{rem} = [[0, 0, -zx-y, 0], 0], G=4$$

$$S(-zx+y, -x^2+z^3y) = -y^2z^2+x^3, \mathit{rem} = [[0, 0, 0, 0], -y^2z^2+x^3], G=4$$

$$S(x-z^4, -y^2z^2+x^3) = -xy^2+z^2x^3, \mathit{rem} = [[0, 0, -zx^2-yx, 0, 0], 0], G=5$$

$$S(y-z^5, -y^2z^2+x^3) = -y^3+z^3x^3, \mathit{rem} = [[0, 0, -z^2x^2-zxy-y^2, 0, 0], 0], G=5$$

$$S(-zx+y, -y^2z^2+x^3) = -y^3z+x^4, \mathit{rem} = [[0, 0, 0, 0, 0], -y^3z+x^4], G=5$$

$$S(-x^2+z^3y, -y^2z^2+x^3) = -x^2y+zx^3, \mathit{rem} = [[0, 0, -x^2, 0, 0], 0], G=5$$

$$S(x-z^4, -y^3z+x^4) = -x^5+z^5y^3, \mathit{rem} = [[-y^3z, 0, 0, 0, 0, -x], 0], G=6$$

$$S(y-z^5, -y^3z+x^4) = -x^4y+z^6y^3, \mathit{rem} = [[-z^2y^3, 0, -y^3z, 0, 0, -y], 0], G=6$$

$$S(-z\,x+y, -y^3\,z+x^4) = -x^3\,y+z^2\,y^3,\ rem = [[0, 0, 0, 0, -y, 0], 0],\ G=6$$

$$S(-x^2+z^3\,y, -y^3\,z+x^4) = -x^6+z^4\,y^4,\ rem = [[-y^4, 0, x\,y^3, 0, 0, -x^2], 0],\ G=6$$

$$S(-y^2\,z^2+x^3, -y^3\,z+x^4) = -x^7+y^5\,z^3,\ rem = [[0, 0, x^2\,y^3, y^4, 0, -x^3], 0],\ G=6$$

$$G := [x-z^4, y-z^5, -z\,x+y, -x^2+z^3\,y, -y^2\,z^2+x^3, -y^3\,z+x^4]$$

```
> GroebnerReduce( G, X, LTgrlex );
```

$$[-x+z^4, z\,x-y, -x^2+z^3\,y, y^2\,z^2-x^3, -y^3\,z+x^4]$$

Scott Cowan
200034814
scottc@sfu.ca

```
>
```

**MATH 800 Assignment 3**
(due June 21, 2006, 9:30)

```
> restart;
```

## 2.8.5(a)

We want to find all critical points of the given function f$(x, y)$. This amounts to solving the system $\{\frac{\partial}{\partial x}f=0, \frac{\partial}{\partial y}f=0\}$. We enter the polynomial function $f$ and compute its partial derivatives.

```
> f := (x^2+y^2-4)*(x^2+y^2-1)+(x-3/2)^2+(y-3/2)^2;
  F := map2( expand@diff, f, [x,y] );
```

$$f := (x^2+y^2-4)(x^2+y^2-1)+\left(x-\frac{3}{2}\right)^2+\left(y-\frac{3}{2}\right)^2$$

$$F := [4x^3+4xy^2-8x-3, 4yx^2+4y^3-8y-3]$$

4

We compute a Groebner basis for the polynomials $\frac{\partial}{\partial x}f, \frac{\partial}{\partial y}f$ w.r.t. lexicographic order. ✓

```
> G := Groebner[Basis]( F, plex(x,y) );
```

$$G := [8y^3-8y-3, -y+x]$$

We take the first element $g_1$ in the Groebner basis, which is a polynomial in $y$ alone, and solve $g_1 = 0$.

```
> y_sols := solve( G[1], {y} );
```

$$y\_sols := \{y=\frac{-1}{2}\}, \{y=\frac{1}{4}+\frac{\sqrt{13}}{4}\}, \{y=\frac{1}{4}-\frac{\sqrt{13}}{4}\}$$ ✓

We get 3 real solutions. From the second polynomial in the Groebner basis we see that the corresponding value of $x$ in each case will be the same value as $y$. So the three solutions are as follows.

```
> seq( {x = subs( sol, y ), y = subs( sol, y )}, sol=[y_sols] );
```

$$\{y=\frac{-1}{2}, x=\frac{-1}{2}\}, \{y=\frac{1}{4}+\frac{\sqrt{13}}{4}, x=\frac{1}{4}+\frac{\sqrt{13}}{4}\}, \{y=\frac{1}{4}-\frac{\sqrt{13}}{4}, x=\frac{1}{4}-\frac{\sqrt{13}}{4}\}$$ ✓

These are the three critical points of the function f$(x, y)$.
We can now verify this with Maple.

```
> solve( F, {x,y} );
  map( allvalues, [%] );
```

$$\{y=\frac{-1}{2}, x=\frac{-1}{2}\},$$

$$\{y=\frac{1}{2}\mathrm{RootOf}(\_Z^2-\_Z-3, label=\_L1), x=\frac{1}{2}\mathrm{RootOf}(\_Z^2-\_Z-3, label=\_L1)\}$$

$$\left[\{y=\frac{-1}{2}, x=\frac{-1}{2}\}, \{y=\frac{1}{4}+\frac{\sqrt{13}}{4}, x=\frac{1}{4}+\frac{\sqrt{13}}{4}\}, \{y=\frac{1}{4}-\frac{\sqrt{13}}{4}, x=\frac{1}{4}-\frac{\sqrt{13}}{4}\}\right]$$

## 2.8.11

### a)

We enter the generators for the ideal $I$ determined by the 3 given equations.

```
> F := [a+b+c-3, a^2+b^2+c^2-5, a^3+b^3+c^3-7];
```

$$F := [a+b+c-3, a^2+b^2+c^2-5, a^3+b^3+c^3-7]$$

We compute a Groebner basis for this ideal.

```
> G := Groebner[Basis]( F, grlex(a,b,c) );
```

$$G := [a+b+c-3, b^2+c^2+b\,c-3\,b-3\,c+2, 3\,c^3-9\,c^2+6\,c+2]$$

Now we need to load the division algorithm components from the Section 2.7 worksheet. We run the division algorithm on the polynomial $a^4+b^4+c^4-9$ divided by the Groebner basis $G$ for $I$.

```
> DIVIDE( a^4+b^4+c^4-9, G, [a,b,c], LTgrlex )[-1];
```

$$0$$

The remainder is 0, which tells us that $a^4+b^4+c^4-9$ is a polynomial combination of the Groebner basis polynomials. But this means that $a^4+b^4+c^4-9$ is in the ideal generated by the Groebner basis $G$, which is the ideal $I$ generated by the original 3 generators. Therefore $a^4+b^4+c^4-9$ is a polynomial combination of the original 3 generators. So it follows from the 3 given equations that the equation $a^4+b^4+c^4=9$ is true.

### b)

We now run the division algorithm on $a^5+b^5+c^5-11$ divided by the Groebner basis $G$.

```
> DIVIDE( a^5+b^5+c^5-11, G, [a,b,c], LTgrlex )[-1];
```

$$\frac{-4}{3}$$

Since the remainder is nonzero, $a^5+b^5+c^5-11$ is not in the ideal $I$. Therefore it follows from the 3 given equations that $a^5+b^5+c^5 \neq 11$.

### c)

To figure out the values of $a^5+b^5+c^5$ and $a^6+b^6+c^6$ we run the division algorithm on these polynomials divided by the Groebner basis $G$.

```
> DIVIDE( a^5+b^5+c^5, G, [a,b,c], LTgrlex )[-1];
  DIVIDE( a^6+b^6+c^6, G, [a,b,c], LTgrlex )[-1];
```

$$\frac{29}{3}$$

$$\frac{19}{3}$$

This tells us that $a^5 + b^5 + c^5 - \frac{29}{3}$ and $a^6 + b^6 + c^6 - \frac{19}{3}$ are polynomial combinations of the polynomials in the Groebner basis *G*, so are in *I*. Therefore it follows from the 3 given equations that $a^5 + b^5 + c^5 = \frac{29}{3}$ and $a^6 + b^6 + c^6 = \frac{19}{3}$.

Scott Cowan
200034814
scottc@sfu.ca

>

**MATH 800 Assignment 3**
(due June 21, 2006, 9:30)

```
> restart;
```

# 3.1.2

## a)

We enter the generators for the ideal $I$ determined by the given equations.

```
> F := [x^2+2*y^2-3, x^2+x*y+y^2-3];
```

$$F := [x^2+2\,y^2-3, x^2+x\,y+y^2-3]$$

We compute Groebner bases for $I$ w.r.t. lex order with $x<y$ and $y<x$.

```
> G_yx := Groebner[Basis]( F, plex(y,x) );
  G_xy := Groebner[Basis]( F, plex(x,y) );
```

$$G\_yx := [x^4-4\,x^2+3, 2\,y+x^3-3\,x]$$

$$G\_xy := [y^3-y, x\,y-y^2, x^2+2\,y^2-3]$$

By the Elimination Theorem, $G\_x = G\_yx \cap k[x]$ is a Groebner basis for $I \cap k[x]$ (the first elimination ideal of $I \subseteq k[y,x]$), and $G\_y = G\_xy \cap k[y]$ is a Groebner basis for $I \cap k[y]$ (the first elimination ideal of $I \subseteq k[x,y]$. We compute $G\_x$ and $G\_y$ by selecting the polynomials from $G\_yx$ in variable $x$ alone and the polynomials from $G\_xy$ in variable $y$ alone, respectively.

2

```
> G_x := map( g-> `if`( indets(g) minus {x} = {}, g, NULL),
  G_yx );
  G_y := map( g-> `if`( indets(g) minus {y} = {}, g, NULL),
  G_xy );
```

$$G\_x := [x^4-4\,x^2+3]$$ ✓

$$G\_y := [y^3-y]$$ ✓

## b)

To solve the equations, we will need to compute the greatest common divisor of an arbitrary number of univariate polynomials. We use this procedure from Assignment #1 (Section 1.5).

```
> GCD := proc() local f, g;
    if (nargs < 1) then return 0 fi;
    if (nargs = 1) then return gcd( args[1], args[1] ) fi;
    return gcd( args[1], GCD( args[2..-1] ) );
  end:
```

We start by computing the partial solutions in $V(I_1) = V(G\_y)$ (where $I_1 = I \cap k[y]$) by setting the only polynomial in $G\_y$ (which happens to be $G\_xy_1$) equal to zero and solving for $y$.

```
> y_sols := [solve( G_xy[1], {y} )];
```

$$y\_sols := [\{y=0\}, \{y=1\}, \{y=-1\}]$$

Note that the leading coefficient of $x$ in $G\_xy_3 = x^2 + 2y^2 - 3$ is a constant, so by the Extension Theorem (or specifically Corollary 4) every partial solution for $y$ extends to a solution for $x, y$. We must simply substitute each partial solution into the remaining Groebner basis polynomials to get univariate polynomials in $x$, find the GCD of these, and solve for where it is zero.

```
> map( subs, y_sols, G_xy[2..-1] );
  map( GCD@op, % );
  x_sols := map( [solve], %, {x} );
```

$$[[0, x^2-3], [x-1, x^2-1], [-x-1, x^2-1]]$$

$$[x^2-3, x-1, x+1]$$

$$x\_sols := [[\{x=\sqrt{3}\}, \{x=-\sqrt{3}\}], [\{x=1\}], [\{x=-1\}]]$$

2

We get 4 solutions over the complex numbers, as summarized below.

```
> sols := seq( seq( seq( {op(xsol), ysol}, xsol=x_sols[i] ),
  ysol=y_sols[i] ), i=1..nops(y_sols) );
```

$$sols := \{y=0, x=\sqrt{3}\}, \{y=0, x=-\sqrt{3}\}, \{y=1, x=1\}, \{y=-1, x=-1\}$$ ✓

We can verify our solutions with Maple.

```
> solve( F, {x,y} );
  map( allvalues, [%] );
```

$$\{y=0, x=\mathrm{RootOf}(\_Z^2-3)\}, \{y=1, x=1\}, \{y=-1, x=-1\}$$

$$[\{y=0, x=\sqrt{3}\}, \{y=0, x=-\sqrt{3}\}, \{y=1, x=1\}, \{y=-1, x=-1\}]$$

### c & d)

We can see that only 2 of these solutions are rational.

```
> sols[3..4];
```

$$\{y=1, x=1\}, \{y=-1, x=-1\}$$ ✓

The smallest field over which all solutions can be expressed is $\mathbb{Q}(\sqrt{3}) = \{a + b\sqrt{3}, a \in Q, b \in Q\}$. ✓

## 3.1.4

We enter the generators for the ideal $I \subseteq k[x, y, z]$ determined by the given equations and compute a Groebner basis $G$ for $I$ w.r.t. lex order.

```
> F := [x^2+y^2+z^2-4, x^2+2*y^2-5, x*z-1];
  G := Groebner[Basis]( F, plex(x,y,z) );
```

$$F := [x^2+y^2+z^2-4, x^2+2y^2-5, xz-1]$$

$$G := [2z^4-3z^2+1, y^2-z^2-1, x+2z^3-3z]$$ ✓

By the Elimination Theorem, $G_1 = G \cap k[y, z]$ is a Groebner basis for $I_1 = I \cap k[y, z]$, and $G_2 = G \cap k[z]$ is a Groebner basis for $I_2 = I \cap k[z]$. We compute $G_1$ and $G_2$ by selecting from $G$ the polynomials in variables $y, z$ and in $z$, respectively. ✓

```
> G1 := map( g-> `if`( indets(g) minus {y,z} = {}, g, NULL), G
  );
  G2 := map( g-> `if`( indets(g) minus {z} = {}, g, NULL), G );
```

$$GI := [2z^4 - 3z^2 + 1, y^2 - z^2 - 1]$$

$$G2 := [2z^4 - 3z^2 + 1]$$

To solve the equations, we start by computing the partial solutions in $V(I_2) = V(G_2)$ for $z$.

```
> z_sols := [solve( G2[-1], {z} )];
```

$$z\_sols := \left[ \{z = -1\}, \{z = 1\}, \{z = \frac{\sqrt{2}}{2}\}, \{z = -\frac{\sqrt{2}}{2}\} \right]$$

Note that the leading coefficient of $y$ in the other polynomial in $G_1$ ($g = y^2 - z^2 - 1$) is a constant, so by the Extension Theorem (Corollary 4) every partial solution for $z$ extends to a partial solution for $y$, $z$ in $V(I_1) = V(G_1)$. We must simply substitute each partial solution for $z$ into $g$ and solve for $y$.

```
> map( subs, z_sols, G1[-1] );
  y_sols := map( [solve], %, {y} );
```

$$\left[ y^2 - 2, y^2 - 2, y^2 - \frac{3}{2}, y^2 - \frac{3}{2} \right]$$

$$y\_sols := \left[ [\{y = \sqrt{2}\}, \{y = -\sqrt{2}\}], [\{y = \sqrt{2}\}, \{y = -\sqrt{2}\}], \left[ \{y = \frac{\sqrt{6}}{2}\}, \{y = -\frac{\sqrt{6}}{2}\} \right], \left[ \{y = \frac{\sqrt{6}}{2}\}, \{y = -\frac{\sqrt{6}}{2}\} \right] \right]$$

Now the leading coefficient of $x$ in the only remaining polynomial in $G$ ($g = x + 2z^3 - 3z$) is a constant, so by the Extension Theorem (Corollary 4) every partial solution for $y$, $z$ extends to a solution for $x$, $y$, $z$ in $V(I) = V(G)$. We must simply substitute each partial solution for $y$, $z$ into $g$ and solve for $x$. Examining $g$ we see that $x$ depends on $z$ alone, so we can actually simplify our work by substituting the partial solutions from $V(I_2)$ for $z$ into $g$.

4

```
> map( subs, z_sols, G[-1] );
  x_sols := map( [solve], %, {x} );
```

$$[x + 1, x - 1, x - \sqrt{2}, x + \sqrt{2}]$$

$$x\_sols := [[\{x = -1\}], [\{x = 1\}], [\{x = \sqrt{2}\}], [\{x = -\sqrt{2}\}]]$$

So for each partial solution for $z$ we get two values for $y$ and one value for $x$. Since there are 4 solutions for $z$, we get a total of 8 solutions in $V(I)$ over the complex numbers, as summarized below.

```
> sols := seq( seq( seq( seq( {op(xsol), op(ysol), zsol},
  xsol=x_sols[i] ), ysol=y_sols[i] ), zsol=z_sols[i] ),
  i=1..nops(z_sols) );
```

$$sols := \{z = -1, x = -1, y = \sqrt{2}\}, \{z = -1, x = -1, y = -\sqrt{2}\}, \{x = 1, z = 1, y = \sqrt{2}\},$$

$\{x = 1, z = 1, y = -\sqrt{2}\}$, $\{z = \frac{\sqrt{2}}{2}, y = \frac{\sqrt{6}}{2}, x = \sqrt{2}\}$, $\{z = \frac{\sqrt{2}}{2}, y = -\frac{\sqrt{6}}{2}, x = \sqrt{2}\}$, $\{x = -\sqrt{2}, z = -\frac{\sqrt{2}}{2}, y = \frac{\sqrt{6}}{2}\}$, $\{x = -\sqrt{2}, z = -\frac{\sqrt{2}}{2}, y = -\frac{\sqrt{6}}{2}\}$

We can verify our solutions with Maple.

```
> solve( F, {x,y,z} );
  map( allvalues, [%] );
```

$$\{x = 1, z = 1, y = \mathrm{RootOf}(\_Z^2 - 2)\}, \{z = -1, x = -1, y = \mathrm{RootOf}(\_Z^2 - 2)\},$$
$$\{x = \mathrm{RootOf}(\_Z^2 - 2), z = \frac{1}{2}\mathrm{RootOf}(\_Z^2 - 2), y = \mathrm{RootOf}(2\_Z^2 - 3)\}$$

$$\Big[\{x = 1, z = 1, y = \sqrt{2}\}, \{x = 1, z = 1, y = -\sqrt{2}\}, \{z = -1, x = -1, y = \sqrt{2}\},$$
$$\{z = -1, x = -1, y = -\sqrt{2}\}, \{z = \frac{\sqrt{2}}{2}, y = \frac{\sqrt{6}}{2}, x = \sqrt{2}\}, \{x = -\sqrt{2}, z = -\frac{\sqrt{2}}{2}, y = \frac{\sqrt{6}}{2}\},$$
$$\{z = \frac{\sqrt{2}}{2}, y = -\frac{\sqrt{6}}{2}, x = \sqrt{2}\}, \{x = -\sqrt{2}, z = -\frac{\sqrt{2}}{2}, y = -\frac{\sqrt{6}}{2}\}\Big]$$

We can see that none of these solutions are rational. These equations have no solutions over $Q$. Note that we could have figured this out with many fewer calculations as follows: There were 2 partial solutions for $z$ that were rational. Any rational solutions must be extended from these. Extending to partial solutions for $y, z$ gives no rational values for $y$. Since the only solutions must be extended from these partial solutions, there can be no rational solutions to the equations for $x, y, z$.

## 3.1.7

### a)

We enter the generators for the ideal $I \subseteq k[t, x, y, z]$ determined by the given equations.

```
> F := [t^2+x^2+y^2+z^2, t^2+2*x^2-x*y-z^2, t+y^3-z^3];
```

$$F := [t^2 + x^2 + y^2 + z^2, t^2 + 2x^2 - xy - z^2, t + y^3 - z^3]$$

We compute a Groebner basis $G$ for $I$ w.r.t. lex order. By the Elimination Theorem, $G_1 = G \cap k[x, y, z]$ is a Groebner basis for $I_1 = I \cap k[x, y, z]$. We compute $G_1$ byselecting from $G$ the polynomials in variables $x, y, z$.

```
> G := Groebner[Basis]( F, plex(t,x,y,z) );
  G1 := map( g-> `if`( indets(g) minus {x,y,z} = {}, g, NULL),
  G );
```

$$G := [y^{12} - 4y^9z^3 + 6z^6y^6 - 4z^9y^3 + z^{12} + 5y^8 + 6y^6z^2 - 10z^3y^5 - 12z^5y^3 + 5z^6y^2$$
$$+ 6z^8 + 5y^4 + 13y^2z^2 + 9z^4, -y^{11} + 4y^8z^3 + xz^6 - 5y^5z^6 + 2z^9y^2 - 5y^7 + 3xz^2$$
$$- 3y^5z^2 + 10z^3y^4 + 6z^5y^2 - 3yz^6 - 5y^3 - 7yz^2, y^6 - 2y^3z^3 + z^6 + xy + 2y^2 + 3z^2,$$

$$x^2+y^6-2y^3z^3+z^6+y^2+z^2,\ t+y^3-z^3]$$

$$GI := [y^{12}-4y^9z^3+6z^6y^6-4z^9y^3+z^{12}+5y^8+6y^6z^2-10z^3y^5-12z^5y^3+5z^6y^2 +6z^8+5y^4+13y^2z^2+9z^4,\ -y^{11}+4y^8z^3+xz^6-5y^5z^6+2z^9y^2-5y^7+3xz^2 -3y^5z^2+10z^3y^4+6z^5y^2-3yz^6-5y^3-7yz^2,\ y^6-2y^3z^3+z^6+xy+2y^2+3z^2, x^2+y^6-2y^3z^3+z^6+y^2+z^2]$$

## b)

We compute a Groebner basis for $I_1 = I \cap k[x, y, z]$ w.r.t. grevlex order, using $G_1$ since we know that this is a basis for $I_1$.

```
> G1_grev := Groebner[Basis]( G1, tdeg(x,y,z) );
```

$$G1\_grev := [x^2-xy-y^2-2z^2,\ y^6-2y^3z^3+z^6+xy+2y^2+3z^2]$$

## c)

We add the given polynomial to our grevlex Groebner basis for $I_1$.

```
> G_elim := [op(G1_grev), t+y^3-z^3];
```

4

$$G\_elim := [x^2-xy-y^2-2z^2,\ y^6-2y^3z^3+z^6+xy+2y^2+3z^2,\ t+y^3-z^3]$$

We want to show that this is a Groebner basis for $I$ w.r.t. the first elimination order defined in Exercise 3.1.6.

We use Maple to compute the reduced Groebner basis for $I$ w.r.t. first elimination order, to show that we get the same result.

```
> Groebner[Basis]( F, lexdeg([t],[x,y,z]) );
```

$$[x^2-xy-y^2-2z^2,\ y^6-2y^3z^3+z^6+xy+2y^2+3z^2,\ t+y^3-z^3]$$


Scott Cowan
200034814
scottc@sfu.ca


```
>
```

**MATH 800 Assignment 3**
(due June 21, 2006, 9:30)

```
> restart;
```

# 3.3.6

## a)

Theorem 1 (for polynomial implicitization) tells us that for the given parametric surface $S$, the ideal $I$ generated by the following polynomials will lead to the smallest variety $V$ that contains $S$.

```
> F := [x-u*v, y-u^2, z-v^2];
```

$$F := [x - u\,v, y - u^2, z - v^2]$$

We compute a Groebner basis for $I$ w.r.t. lex order, along with the bases (by the Elimination Theorem) for the first and second elimination ideals of $I$. Then (Theorem 1 gives) $V = \mathrm{V}(I_2)$.

```
> G := Groebner[Basis]( F, plex(u,v,x,y,z) );
  G1 := map( g-> `if`( indets(g) minus {v,x,y,z} = {}, g,
  NULL), G );
  G2 := map( g-> `if`( indets(g) minus {x,y,z} = {}, g, NULL),
  G1 );
```

$$G := [x^2 - y\,z, -z + v^2, u\,z - v\,x, u\,x - v\,y, -x + u\,v, -y + u^2]$$

$$G1 := [x^2 - y\,z, -z + v^2]$$

$$G2 := [x^2 - y\,z]$$ ✓

So the equation of $V$, given by the single polynomial in $G_2$, is $x^2 - y\,z = 0$, or $x^2 = y\,z$.

## b)

To show that $V = S$ over the complex numbers $C$, i.e. that $S$ fills up all of $V$, we need to show that all partial solutions for $[x, y, z] \in \mathrm{V}(I_2)$ extend to solutions for $[u, v, x, y, z] \in \mathrm{V}(I)$.

```
> lcoeff( G1[2], v );
```

$$1$$

The leading coefficient of $v$ in one of the polynomials in $G_1$ is constant, so by the Extension Theorem every partial solution in $\mathrm{V}(I_2)$ extends to a partial solution for $[v, x, y, z] \in \mathrm{V}(I_1)$.

```
> lcoeff( G[6], u );
```

$$1$$

Then the leading coefficient of $u$ in one of the polynomials in $G$ is constant, so by the Extension Theorem every partial solution in $\mathrm{V}(I_1)$ extends to a partial solution for $[u, v, x, y, z] \in \mathrm{V}(I)$. Therefore $V = S$. ✓

## c)

Over the real numbers $R$, the extension from $\mathrm{V}(I_2)$ to $\mathrm{V}(I_1)$ will fail if $z$ takes a negative value, because

```
> G1[2];
```

$$-z+v^2$$

will have no real roots for $v$. Similarly the extension from $V(I_1)$ to $V(I)$ will fail if $y$ takes a negative value, because

```
> G[6];
```

$$-y+u^2$$

will have no real roots for $u$. So the parametrization defining $S$ only covers $y, z$ nonnegative, while the implicit equation for $V$ allows $y, z$ to be negative. Graphically, we get the following images from plotting the implicit equation for $V$ and the parametric equations for $S$.

```
> plots[implicitplot3d]( G2, x=-4..4, y=-4..4, z=-4..4,
  grid=[25,25,25], style=patchcontour, axes=boxed,
  orientation=[30,60] );
```



```
> plot3d( [s*t,s^2,t^2], s=-2..2, t=-2..2, style=patchcontour,
  axes=boxed, labels=[x,y,z], view=[-4..4,-4..4,-4..4],
  orientation=[30,60] );
```




+1 bonus.

We see quite clearly that the parametrization gives only half of the variety. To parametrize the other half, we must allow $y$, $z$ to be negative (simultaneously, it turns out) but still satisfying the equation $x^2 - y\,z = 0$ of $V$. This can be done with the parametrization $x = u\,v, y = -u^2, z = -v^2$. Plotting the original parametric surface with this new one, we see that this gives the entire variety over $R$.

```
> plot3d( [[s*t,s^2,t^2], [s*t,-s^2,-t^2]], s=-2..2, t=-2..2,
  style=patchcontour, axes=boxed, labels=[x,y,z],
  view=[-4..4,-4..4,-4..4], orientation=[30,60] );
```




6+1

## 3.3.8

### a)

Theorem 1 (for polynomial implicitization) tells us that for the given parametric surface $S$ (the Enneper surface), the ideal $I$ generated by the following polynomials will lead to the smallest variety $V$ that contains $S$.

```
> F := [x-3*u-3*u*v^2+u^3, y-3*v-3*u^2*v+v^3, z-3*u^2+3*v^2];
```

$$F := [x - 3\,u - 3\,v^2\,u + u^3, y - 3\,v - 3\,v\,u^2 + v^3, z - 3\,u^2 + 3\,v^2]$$

We compute a Groebner basis for $I$ w.r.t. lex order, along with the bases (by the Elimination Theorem) for the first and second elimination ideals of $I$. Then (Theorem 1 gives) $V = \mathrm{V}(I_2)$.

```
> G := Groebner[Basis]( F, plex(u,v,x,y,z) ):
```

```
G1 := map( g-> `if`( indets(g) minus {v,x,y,z} = {}, g,
NULL), G ):
G2 := map( g-> `if`( indets(g) minus {x,y,z} = {}, g, NULL),
G1 );
```

$$G2 := [-64 z^9 + 1296 z^6 x^2 - 1296 z^6 y^2 + 10935 z^3 x^4 + 56862 z^3 x^2 y^2 + 34992 z^5 x^2 + 10935 y^4 z^3 + 34992 z^5 y^2 + 10368 z^7 + 19683 x^6 - 59049 x^4 y^2 + 118098 z^2 x^4 + 59049 y^4 x^2 + 174960 z^4 x^2 - 19683 y^6 - 118098 z^2 y^4 - 174960 z^4 y^2 - 59049 z x^4 + 118098 z x^2 y^2 - 314928 x^2 z^3 - 59049 z y^4 - 314928 y^2 z^3 - 419904 z^5]$$

```
> nops(G), nops(G1), nops(G2);
```

$$15, 9, 1$$

Note that $G$ contains 15 complicated polynomials, and took a noticable delay to compute. So the equation of $V$ containing the Enneper surface is given by the single polynomial in $G_2$ (set equal to zero).

## b)

To show that $V = S$ over the complex numbers $C$, i.e. that $S$ fills up all of $V$, we use the Extension Theorem. We compute the leading coefficients for $v$ of the polynomials in $G_1$ and look at their variables.

```
> map(indets@lcoeff, G1, v);
  `intersect`(op(%));
```

$$[\{x, y, z\}, \{y, z\}, \{x, y, z\}, \{x, y, z\}, \{x, y, z\}, \{z\}, \{y, z\}, \{x, y, z\}, \{\}]$$

$$\{\}$$

This tells us that the last polynomial in $G_1$ has a constant leading coefficient for $v$. We look at this polynomial and leading coefficient.

```
> G1[-1], lcoeff(G1[-1], v);
```

$$-y + 3v + 2v^3 + vz, 2$$

So every partial solution $[x, y, z] \in \mathrm{V}(I_2)$ extends to $[v, x, y, z] \in \mathrm{V}(I_1)$.
Then we compute the leading coefficients for $u$ of the polynomials in $G$ and look at their variables.

3

```
> map(indets@lcoeff, G, u);
  `intersect`(op(%));
```

$$[\{x, y, z\}, \{v, x, y, z\}, \{v, x, y, z\}, \{v, x, y, z\}, \{v, x, y, z\}, \{v, x, y, z\}, \{v, x, y, z\}, \{v, x, y, z\}, \{v, y, z\}, \{y, z\}, \{x\}, \{v, y, z\}, \{v, y, z\}, \{v, z\}, \{\}]$$

$$\{\}$$

Here the last polynomial in $G$ has a constant leading coefficient for $u$. We look at this polynomial and leading coefficient.

```
> G[-1], lcoeff(G[-1], u);
```

$$-z + 3u^2 - 3v^2, 3$$

So every partial solution $[v, x, y, z] \in \mathrm{V}(I_1)$ extends to $[u, v, x, y, z] \in \mathrm{V}(I)$. This proves that $V = S$.

We show what the Enneper surface looks like, both from its implicit equation and from its parametrization. This also helps convince us that the parametric surface $S$ equals the variety $V$ over $R$ as well as over $C$.

```
> plots[implicitplot3d]( G2, x=-10..10, y=-10..10, z=-10..10,
  grid=[25,25,25], style=patchcontour, axes=frame,
  orientation=[45,30] );
```

+1 Bonus.




```
> plot3d( [3*s+3*s*t^2-s^3, 3*t+3*s^2*t-t^3, 3*s^2-3*t^2],
  s=-2..2, t=-2..2, style=patchcontour, axes=frame,
  labels=[x,y,z], view=[-10..10,-10..10,-10..10],
  orientation=[45,30] );
```




## 3.3.14

### a)

Theorem 2 (for rational implicitization) tells us that for the given parametric curve (the folium of Descartes), the ideal $I$ generated by the following polynomials will lead to the smallest variety $V$ that contains the curve.

```
> F := [expand((1+t^3)*x-3*t), expand((1+t^3)*y-3*t^2),
  expand(1-(1+t^3)*(1+t^3)*u)];
```

$$F := [x + x\,t^3 - 3\,t, y + y\,t^3 - 3\,t^2, 1 - u - 2\,u\,t^3 - u\,t^6]$$

However Exercise 13 implies that we don't need the third polynomial. This is because the denominator $1 + t^3$ is relatively prime with the numerators $3\,t$, $3\,t^2$, so there is no way that the first two polynomials can vanish without the third automatically vanishing. So we redefine $F$.

```
> F := F[1..2];
```

$$F := [x + x\,t^3 - 3\,t, y + y\,t^3 - 3\,t^2]$$

We compute a Groebner basis for $I$ w.r.t. lex order, along with the basis (by the Elimination Theorem) for the first elimination ideal of $I$. Then (Theorem 1 gives) $V = \mathrm{V}(I_1)$.

```
> G := Groebner[Basis]( F, plex(t,x,y) );
  G1 := map( g-> `if`( indets(g) minus {x,y} = {}, g, NULL), G
  );
```

$$G := [y^3 + x^3 - 3\,y\,x, y^2\,t + x^2 - 3\,y, -y + x\,t, x - 3\,t + y\,t^2]$$

$$G1 := [y^3 + x^3 - 3\,y\,x]$$

So the equation of $V$, given by the single polynomial in $G_1$, is $x^3 + y^3 - x\,y = 0$.

**b)**

4

To show that $V$ is the folium over the complex numbers $C$, i.e. that the parametrization fills up all of $V$, we use the Extension Theorem. We compute the leading coefficients for $t$ of the polynomials in $G$.

```
> map( lcoeff, G, t );
```

$$[y^3 + x^3 - 3\,y\,x, y^2, x, y]$$

Now these polynomials in $x, y$ all vanish if $x = 0, y = 0$, and don't all vanish if one of $x, y$ is nonzero (which can be seen from the last two). Their variety is $\{[0, 0]\}$, a single point. We can verify this with Maple's solving procedure.

```
> solve( %, {x,y} );
```

$$\{x = 0, y = 0\}$$

What this tells us is that any partial solution $[x, y] \neq [0, 0]$ must extend (by the Extension Theorem) to a solution $[t, x, y] \in \mathrm{V}(I)$. So if $[x, y] = [0, 0]$ is a partial solution then it is the only one that might not extend to a solution $[t, x, y] \in \mathrm{V}(I)$. We also see that $[x, y] = [0, 0]$ is in fact a partial solution. We investigate this partial solution further.

```
> subs( {x=0,y=0}, G );
  solve( %, {t} );
```

$$[0, 0, 0, -3\,t]$$

$$\{t = 0\}$$

By evaluating the basis $G$ for $I$ at $[x, y] = [0, 0]$ we see that this partial solution can indeed be extended with $t = 0$. Thus every partial solution $[x, y] \in \mathrm{V}(I_1)$ extends to $[t, x, y] \in \mathrm{V}(I)$ over $C$.

To show that the parametrization covers the entire curve over $R$ requires more work. Let us again list $G$ for reference.

```
> G;
```

$$[y^3 + x^3 - 3\,y\,x, y^2\,t + x^2 - 3\,y, -y + x\,t, x - 3\,t + y\,t^2]$$

First we consider the partial solution $[x, y] = [0, 0]$. It follows exactly as it did for $C$ that this extends over $R$ with $t = 0$.

Next we consider any other partial solution with $x = 0$.

```
> subs( x=0, G );
```

$$[y^3, y^2 t - 3y, -y, y t^2 - 3t]$$

It turns out that for $G$ to vanish $y$ must be zero. So there are no partial solutions with $x = 0, y \neq 0$.

The final case is for partial solutions with $x \neq 0, y \neq 0$. As the polynomials in $G$ are given, we do not immediately see a common solution for $t$ that will cause them to all vanish. However we can add and subtract multiples of the first polynomial to the others without changing any solutions for $t$. This is because the first polynomial vanishes on the partial solution for $x$, $y$, and because neither $x$ nor $y$ can be zero.

```
> expand(x*G[2]-G[1]), G[3], expand(x*y*G[4]-t*G[1]);
  solve( {%}, {t} );
```

$$-y^3 + x y^2 t, -y + x t, y x^2 + x t^2 y^2 - t y^3 - t x^3$$

$$\{t = \frac{y}{x}\}$$

We see that there is a common solution $t = \frac{y}{x}$, which is always defined since $x \neq 0$, so these partial solutions for $x$, $y$ extend over $R$ too.

We finish by showing the folium of Descartes, both from its implicit equation and from its parametrization.

```
> plots[implicitplot]( G1, x=-3..3, y=-3..3, grid=[100,100],
  scaling=constrained );
```




Scott Cowan
200034814
scottc@sfu.ca

>

MATH 800 Assignment 3
(due June 21, 2006, 9:30)

This worksheet uses Buchberger's algorithm components from the Section 2.7 worksheet.

# Additional Exercise 1

We enter the generators $f_1, f_2$. We will use lex order throughout this exercise.

```
> (f1,f2) := (x^2*y-z,x*y-1);
```

$$f1, f2 := x^2\, y - z, x\, y - 1$$

We enter $r$ and $g$ that we found in Exercise 2.3.5.

```
> r := f1+expand(-x*f2);
  g := expand(-y*f1)+expand((x*y+1)*f2);
```

$$r := -z + x$$

$$g := y\, z - 1$$

We show how $r$ and $g$ have non-zero remainders on division by $[f_1, f_2]$.

```
> DIVIDE( r, [f1,f2], X, LTlex );
  DIVIDE( g, [f1,f2], X, LTlex );
```

$$[[0, 0], -z + x]$$

$$[[0, 0], y\, z - 1]$$

✓

We compute a reduced Groebner basis for the ideal generated by $f_1, f_2$.

```
> G := Buchberger( [f1,f2], X, LTlex, verbose=true );
  G := GroebnerReduce( G, X, LTlex, verbose=true );
```

$$\mathrm{S}(x^2\, y - z, x\, y - 1) = -z + x, \mathit{rem} = [[0, 0], -z + x], G = 2$$

$$\mathrm{S}(x^2\, y - z, -z + x) = -z + x\, y\, z, \mathit{rem} = [[0, z, 0], 0], G = 3$$

$$\mathrm{S}(x\, y - 1, -z + x) = y\, z - 1, \mathit{rem} = [[0, 0, 0], y\, z - 1], G = 3$$

$$\mathrm{S}(x^2\, y - z, y\, z - 1) = -z^2 + x^2, \mathit{rem} = [[0, 0, x + z, 0], 0], G = 4$$

$$\mathrm{S}(x\, y - 1, y\, z - 1) = -z + x, \mathit{rem} = [[0, 0, 1, 0], 0], G = 4$$

$$\mathrm{S}(-z + x, y\, z - 1) = -y\, z^2 + x, \mathit{rem} = [[0, 0, 1, -z], 0], G = 4$$

$$G := [x^2\, y - z, x\, y - 1, -z + x, y\, z - 1]$$

$$g = x^2\, y - z, \mathit{rem} = [[x, 1, 0], 0], G = 3$$

$$g = x\, y - 1, \mathit{rem} = [[y, 1], 0], G = 2$$

$$g = -z + x, \mathit{rem} = [[0], -z + x], G = 1$$

$$g = y\, z - 1, \mathit{rem} = [[0], y\, z - 1], G = 1$$

$$G := [-z + x, y\, z - 1]$$

3

We compute the remainders of $r$ and $g$ on division by the Groebner basis $G$.

```
> DIVIDE( r, G, X, LTlex );
  DIVIDE( g, G, X, LTlex );
```

✓

$$[[1, 0], 0]$$

$$[[0, 1], 0]$$

The remainders are both zero. This isn't surprising, since $r$ and $g$ are the polynomials in the Groebner basis!

## Additional Exercise 2

We input the generating sets for the four ideals.

```
> Fa := [y^3-z^2, x*z-y^2, x*y-z, x^2-y];
  Fb := [x*y-z^2, x*z-y^2, x*y-z, x^2-y];
  Fc := [x*z-y^2, x+y^2-z-1, x*y*z-1];
  Fd := [y^2-x^2*y, z-x*y, y-x^2];
```

$$Fa := [y^3 - z^2, x\,z - y^2, x\,y - z, x^2 - y]$$
$$Fb := [x\,y - z^2, x\,z - y^2, x\,y - z, x^2 - y]$$
$$Fc := [x\,z - y^2, x + y^2 - z - 1, x\,y\,z - 1]$$
$$Fd := [y^2 - x^2\,y, z - x\,y, y - x^2]$$

We compute a reduced Groebner basis for each.

3

```
> Ga := {op(Groebner[Basis]( Fa, plex(x,y,z) ))};
  Gb := {op(Groebner[Basis]( Fb, plex(x,y,z) ))};
  Gc := {op(Groebner[Basis]( Fc, plex(x,y,z) ))};
  Gd := {op(Groebner[Basis]( Fd, plex(x,y,z) ))};
```

$$Ga := \{x\,y - z, x^2 - y, y^3 - z^2, x\,z - y^2\}$$
$$Gb := \{x\,y - z, x^2 - y, x\,z - y^2, -z + z^2, y^3 - z, y^2\,z - y^2\}$$
$$Gc := \{x + y^2 - z - 1, -z^3 + y\,z - z^2 + y, z^4 + z^3 - z - 1, y^3 - 1\}$$
$$Gd := \{x\,y - z, x^2 - y, y^3 - z^2, x\,z - y^2\}$$

Recall that every ideal has a unique reduced Groebner basis w.r.t. a fixed monomial ordering. We see that of the four Groebner bases computed, only the first and the last are the same. Therefore the ideals in (a) and (d) are the same, but the other two are different from these and from each other.

Scott Cowan
200034814
scottc@sfu.ca

```
>
```